
1. Overview

This document lists all the essencial changes needed to correct problems in the TTCN implementation of test case 22.4.15 which is part of the NBIOT test suite in the ‘iwd-TTCN3-B2016-06_D17wk23’ ATS delivery. 

With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence. 

Contact:
Yongsheng MA

mayongsheng@starpointcomm.com

Tel. +8615117939410

1.1 Verification Test Summary

Test Case:
TC_22_4_15
ATS Version:
iwd-TTCN3-B2016-06_D17wk23
System Simulator used:
Starpoint SP8630

UE used:
RDA8909
Verification Status:
PASS
2. Corrections required

2.1 Change 1 – Correction to function ‘function f_NAS_GetExtdProtocolConfigOptionList ()’

	Function name
	f_NAS_GetExtdProtocolConfigOptionList ()

	Reason for change
	When UE transmited NAS message PDN CONNECTIVITY REQUEST with esmInfoTransferFlag and then sent ESM INFORMATION RESPONSE, if extdProtocolConfigurationOptions. pco. protocolID = '8021'O in ESM INFORMATION RESPONSE, the v_PcoToUE.pco will be {} returned by f_NAS_GetExtdProtocolConfigOptionList and f_GetDefaultExtdProtocolConfigOptions, so extdProtocolConfigurationOptions in ACTIVATE DEFAULT EPS BEARER CONTEXT REQUEST will be not initialized.

	Summary of change
	1. Initialize v_ProtocolContainerList to omit instead of {} and change its template type to omit.

2. Change template type of return value to no restrict.

	TTCN module
	CIOT_NASTemplates.ttcn

	MCC160 Comment
	


Before change

  function f_NAS_GetExtdProtocolConfigOptionList(NAS_ExtdProtocolConfigOptions_Type p_ConfigOptionsRX,

                                                 boolean p_IgnoreIM_CN_SubsystemSignalingFlag := true) return template (value) NAS_ExtdProtocolConfigOptions_Type

  {

    var PDN_AddressInfo_Type v_PDN_AddressInfo := f_PDN_AddressInfo_Get(PDN_1);

    var template (value) NAS_ExtdProtocolConfigOptions_Type v_ProtocolContainerList := {};

    var integer v_PcoCnt := 0;

    var O2_Type v_ProtocolId;

    var octetstring v_Contents;

    var integer i;

    for (i := 0; i < lengthof(p_ConfigOptionsRX); i := i + 1) {

      v_ProtocolId := p_ConfigOptionsRX[i].protocolID;

      v_Contents := ''O;

      select (v_ProtocolId) {  // See 24.008 Table 10.5.154

        case ('0002'O) {            // IM CN Subsystem Signalling Flag, reply with an empty container

          if (not p_IgnoreIM_CN_SubsystemSignalingFlag) {

            v_ProtocolContainerList[v_PcoCnt] := cs_ProtocolContainer_Common(v_ProtocolId);

            v_PcoCnt := v_PcoCnt + 1;

            continue;

          }

        }

        case ('0003'O) {       // DNS Server IPv6 address

          v_Contents := f_Convert_IPv6Addr2OctString(v_PDN_AddressInfo.DNS_ServerAddressIPv6);

        }

        case ('000D'O) {       // DNS Server IPv4 address

          v_Contents := f_Convert_IPv4Addr2OctString(v_PDN_AddressInfo.DNS_ServerAddressIPv4);

        }

      }

      if (lengthof(v_Contents) > 0) {

        v_ProtocolContainerList[v_PcoCnt] := cs_ProtocolContainer(v_ProtocolId, v_Contents);

        v_PcoCnt := v_PcoCnt + 1;

      }

    }

    return v_ProtocolContainerList;

  }
After change

  function f_NAS_GetExtdProtocolConfigOptionList(NAS_ExtdProtocolConfigOptions_Type p_ConfigOptionsRX,

                                                 boolean p_IgnoreIM_CN_SubsystemSignalingFlag := true) return template (omit) NAS_ExtdProtocolConfigOptions_Type

  {

    var PDN_AddressInfo_Type v_PDN_AddressInfo := f_PDN_AddressInfo_Get(PDN_1);

    var template (omit) NAS_ExtdProtocolConfigOptions_Type v_ProtocolContainerList := omit;

    var integer v_PcoCnt := 0;

    var O2_Type v_ProtocolId;

    var octetstring v_Contents;

    var integer i;

    for (i := 0; i < lengthof(p_ConfigOptionsRX); i := i + 1) {

      v_ProtocolId := p_ConfigOptionsRX[i].protocolID;

      v_Contents := ''O;

      select (v_ProtocolId) {  // See 24.008 Table 10.5.154

        case ('0002'O) {            // IM CN Subsystem Signalling Flag, reply with an empty container

          if (not p_IgnoreIM_CN_SubsystemSignalingFlag) {

            v_ProtocolContainerList[v_PcoCnt] := cs_ProtocolContainer_Common(v_ProtocolId);

            v_PcoCnt := v_PcoCnt + 1;

            continue;

          }

        }

        case ('0003'O) {       // DNS Server IPv6 address

          v_Contents := f_Convert_IPv6Addr2OctString(v_PDN_AddressInfo.DNS_ServerAddressIPv6);

        }

        case ('000D'O) {       // DNS Server IPv4 address

          v_Contents := f_Convert_IPv4Addr2OctString(v_PDN_AddressInfo.DNS_ServerAddressIPv4);

        }

      }

      if (lengthof(v_Contents) > 0) {

        v_ProtocolContainerList[v_PcoCnt] := cs_ProtocolContainer(v_ProtocolId, v_Contents);

        v_PcoCnt := v_PcoCnt + 1;

      }

    }

    return v_ProtocolContainerList;

  }
2.2 Change 2 – Correction to function ‘function fl_NASEMU_UL_Handler_NBIOT ()’

	Function name
	fl_NASEMU_UL_Handler_NBIOT ()

	Reason for change
	Correct handling for rrcConnectionResumeComplete and add the handling for no NAS message situation.

	Summary of change
	Correct handling for rrcConnectionResumeComplete and add the handling for no NAS message situation.

	TTCN module
	NasEmu_Main_NBIOT.ttcn

	MCC160 Comment
	


Before change

  function fl_NASEMU_UL_Handler_NBIOT(inout NasSecurity_Type p_NasSecurityByRef,

                                      template (value) NB_IndAspCommonPart_Type p_IndAspCommonPart,

                                      NB_RRC_MSG_Indication_Type p_RrcIndication) runs on NASEMU_NBIOT_PTC

  {

……
    v_SRB_COMMON_IND := cas_NBIOT_SignallingAspToTC_Init(p_IndAspCommonPart);   // RRC and NAS both are initialised as omit

    if (ischosen(p_RrcIndication.Ccch)) {                                     /* CCCH -> SRB0 */

      v_SRB_COMMON_IND.Signalling.Rrc := p_RrcIndication;

    } else {                                                                  /* DCCH -> SRB1/2 */

      v_RrcPdu := p_RrcIndication.Dcch;

      v_RrcPduPresent := true;

      v_NoOfNasPdus := 0;

      v_IntegrityError := false;

      if (ischosen(v_RrcPdu.message_.c1)) {

        if (ischosen(v_RrcPdu.message_.c1.rrcConnectionSetupComplete_r13)) {

          // RRC CONNECTION SETUP COMPLETE

          if (ischosen(v_RrcPdu.message_.c1.rrcConnectionSetupComplete_r13.criticalExtensions.rrcConnectionSetupComplete_r13)) {

            v_NAS_DedicatedInformation[0] := v_RrcPdu.message_.c1.rrcConnectionSetupComplete_r13.criticalExtensions.rrcConnectionSetupComplete_r13.dedicatedInfoNAS_r13;

            v_NoOfNasPdus := 1;

          }

        } else if (ischosen(v_RrcPdu.message_.c1.ulInformationTransfer_r13)) {

          // UL INFORMATION TRANSFER

          if (ischosen(v_RrcPdu.message_.c1.ulInformationTransfer_r13.criticalExtensions.ulInformationTransfer_r13)) {

            v_ULInformationTransfer := v_RrcPdu.message_.c1.ulInformationTransfer_r13.criticalExtensions.ulInformationTransfer_r13;

            v_NAS_DedicatedInformation[0] := v_ULInformationTransfer.dedicatedInfoNAS_r13;

            v_RrcPduPresent := false;

            v_NoOfNasPdus := 1;

          }

        } else {

          // OTHER RRC PDU

        }

      } else if (ischosen(v_RrcPdu.message_.c1.rrcConnectionResumeComplete_r13)) {

        v_NAS_DedicatedInformation[0] := v_RrcPdu.message_.c1.rrcConnectionResumeComplete_r13.criticalExtensions.rrcConnectionResumeComplete_r13.dedicatedInfoNAS_r13;

        v_NoOfNasPdus := 1;

      }
……
    }

    TC_SRB.send(v_SRB_COMMON_IND);

  }
After change

  function fl_NASEMU_UL_Handler_NBIOT(inout NasSecurity_Type p_NasSecurityByRef,

                                      template (value) NB_IndAspCommonPart_Type p_IndAspCommonPart,

                                      NB_RRC_MSG_Indication_Type p_RrcIndication) runs on NASEMU_NBIOT_PTC

  {

    ……
    v_SRB_COMMON_IND := cas_NBIOT_SignallingAspToTC_Init(p_IndAspCommonPart);   // RRC and NAS both are initialised as omit

    if (ischosen(p_RrcIndication.Ccch)) {                                     /* CCCH -> SRB0 */

      v_SRB_COMMON_IND.Signalling.Rrc := p_RrcIndication;

    } else {                                                                  /* DCCH -> SRB1/2 */

      v_RrcPdu := p_RrcIndication.Dcch;

      v_RrcPduPresent := true;

      v_NoOfNasPdus := 0;

      v_IntegrityError := false;

      if (ischosen(v_RrcPdu.message_.c1)) {

        if (ischosen(v_RrcPdu.message_.c1.rrcConnectionSetupComplete_r13)) {

          // RRC CONNECTION SETUP COMPLETE

          if (ischosen(v_RrcPdu.message_.c1.rrcConnectionSetupComplete_r13.criticalExtensions.rrcConnectionSetupComplete_r13)) {

            v_NAS_DedicatedInformation[0] := v_RrcPdu.message_.c1.rrcConnectionSetupComplete_r13.criticalExtensions.rrcConnectionSetupComplete_r13.dedicatedInfoNAS_r13;

            v_NoOfNasPdus := 1;

          }

        } else if (ischosen(v_RrcPdu.message_.c1.ulInformationTransfer_r13)) {

          // UL INFORMATION TRANSFER

          if (ischosen(v_RrcPdu.message_.c1.ulInformationTransfer_r13.criticalExtensions.ulInformationTransfer_r13)) {

            v_ULInformationTransfer := v_RrcPdu.message_.c1.ulInformationTransfer_r13.criticalExtensions.ulInformationTransfer_r13;

            v_NAS_DedicatedInformation[0] := v_ULInformationTransfer.dedicatedInfoNAS_r13;

            v_RrcPduPresent := false;

            v_NoOfNasPdus := 1;

          }

        } else if (ischosen(v_RrcPdu.message_.c1.rrcConnectionResumeComplete_r13)) {

if(ispresent(v_RrcPdu.message_.c1.rrcConnectionResumeComplete_r13.criticalExtensions.rrcConnectionResumeComplete_r13.dedicatedInfoNAS_r13)){

            v_NAS_DedicatedInformation[0] := v_RrcPdu.message_.c1.rrcConnectionResumeComplete_r13.criticalExtensions.rrcConnectionResumeComplete_r13.dedicatedInfoNAS_r13;

            v_NoOfNasPdus := 1;

          }

        }else {

          // OTHER RRC PDU

        }
      }
      ……
    }

    TC_SRB.send(v_SRB_COMMON_IND);

  }
2.3 Change 3 – Correction to function ‘function f_NBIOT_RRCConnectionResume()’

	Function name
	f_NBIOT_RRCConnectionResume()

	Reason for change
	1.  Incorrect Security Header value for NAS message TAU REQUEST in Resume procedure, TAU REQ should be with tsc_SHT_IntegrityProtected. Add new parameter p_ExpectedSecurityStatus instead of tsc_SHT_IntegrityProtected_Ciphered.
2.  Add default Continuous UL Grant handling in Resume procedure.
3.  Change type of return value v_NAS_UL_Message to var template instead of var, so Msg4 configuration with RRCConnectionSetup can be set to SS before  v_NAS_UL_Message is returned.

	Summary of change
	1.  Add new parameter p_ExpectedSecurityStatus instead of tsc_SHT_IntegrityProtected_Ciphered.
2.  Add default Continuous UL Grant handling in Resume procedure. 

3.  Change type of return value v_NAS_UL_Message to var template instead of var, so Msg4 configuration with RRCConnectionSetup can be set to SS before  v_NAS_UL_Message is returned.

	TTCN module
	NBIOT_RRCSteps.ttcn

	MCC160 Comment
	


Before change

  function f_NBIOT_RRCConnectionResume(NBIOT_CellId_Type p_CellId,

                                       template (present) EUTRA_ASN1_ResumeIdentity_r13_Type p_ResumeIdentity := ?,

                                       template (present) EUTRA_ASN1_ShortMAC_I_Type p_ShortMAC_I := ?,

                                       template (present) EstablishmentCause_NB_r13 p_ResumeCause := ?,

                                       template NAS_UL_Message_Type p_Expected_NAS_UL_Message := *) runs on NBIOT_PTC return template (omit) NAS_UL_Message_Type

  { /* Steps 1 - 3 of procedure 8.1.5A.6 but with optional NAS messages (being TAU for 8.1.5A.6 but omit in general) */

    var EUTRA_ASN1_RRC_TransactionIdentifier_Type v_RRC_TI := tsc_RRC_TI_Def;

    var template (value) DRB_ToAddModList_NB_r13 v_DRB_ToAddModList := { cs_508_NB_DRB_ToAddMod_DEFAULT(tsc_DRB1) };

    var template (value) RadioResourceConfigDedicated_NB_r13 v_RadioResourceConfigDedicated := cs_RadioResourceConfigDedicated_NB_DrbConfig(v_DRB_ToAddModList);

    var NBIOT_SecurityParams_Type v_SecurityParams;

    var NB_SRB_COMMON_IND v_NB_SRB_COMMON_IND;

    var NAS_UL_Message_Type v_NAS_UL_Message;

    SRB.receive(car_NB_SRB0_RrcPdu_IND(p_CellId, cr_RRCConnectionResumeRequest_NB(p_ResumeIdentity, p_ShortMAC_I, p_ResumeCause)));

    v_SecurityParams := f_NBIOT_Security_Get();

    v_SecurityParams := f_NBIOT_SS_AS_ReestablishResumeSecurity(p_CellId, v_SecurityParams, rrcConnectionResume);

    f_NBIOT_Security_Set(v_SecurityParams);

    SRB.send(cas_NB_SRB_RrcPdu_REQ(p_CellId, tsc_SRB1, cs_TimingInfo_Now, cs_RRCConnectionResume_NB(v_RRC_TI, v_RadioResourceConfigDedicated)));

    alt {

      [] SRB.receive(car_NB_SRB_RrcNasPdu_IND(p_CellId, tsc_SRB1, cr_RRCConnectionResumeComplete_NB(v_RRC_TI), cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered, ?))) -> value v_NB_SRB_COMMON_IND {

        v_NAS_UL_Message := v_NB_SRB_COMMON_IND.Signalling.Nas[0].Pdu.Msg;

        if (not match(v_NAS_UL_Message, p_Expected_NAS_UL_Message)) {

          f_NBIOT_SetVerdictFailOrInconc (__FILE__, __LINE__, "RRCConnectionResumeComplete: Piggybacked NAS message does not match");

        }

        return v_NAS_UL_Message;
      }

      [] SRB.receive(car_NB_SRB_RrcPdu_IND(p_CellId, tsc_SRB1, cr_RRCConnectionResumeComplete_NB(v_RRC_TI))) {

        if (ispresent(p_Expected_NAS_UL_Message)) {

          f_NBIOT_SetVerdictFailOrInconc (__FILE__, __LINE__, "RRCConnectionResumeComplete: Piggybacked NAS message expected");

        }

      }

    }

    f_NBIOT_SS_ConfigRachProcedureMsg4(p_CellId, f_NBIOT_508_RRCConnectionSetup());

    return omit;

  }
After change

  function f_NBIOT_RRCConnectionResume(NBIOT_CellId_Type p_CellId,

                                       template (present) EUTRA_ASN1_ResumeIdentity_r13_Type p_ResumeIdentity := ?,

                                       template (present) EUTRA_ASN1_ShortMAC_I_Type p_ShortMAC_I := ?,

                                       template (present) EstablishmentCause_NB_r13 p_ResumeCause := ?,

                                       template NAS_UL_Message_Type p_Expected_NAS_UL_Message := *,










          template (present) SecurityHeaderType p_ExpectedSecurityStatus := tsc_SHT_IntegrityProtected_Ciphered ) runs on NBIOT_PTC return template (omit) NAS_UL_Message_Type

  { /* Steps 1 - 3 of procedure 8.1.5A.6 but with optional NAS messages (being TAU for 8.1.5A.6 but omit in general) */

    var EUTRA_ASN1_RRC_TransactionIdentifier_Type v_RRC_TI := tsc_RRC_TI_Def;

    var template (value) DRB_ToAddModList_NB_r13 v_DRB_ToAddModList := { cs_508_NB_DRB_ToAddMod_DEFAULT(tsc_DRB1) };

    var template (value) RadioResourceConfigDedicated_NB_r13 v_RadioResourceConfigDedicated := cs_RadioResourceConfigDedicated_NB_DrbConfig(v_DRB_ToAddModList);

    var NBIOT_SecurityParams_Type v_SecurityParams;

    var NB_SRB_COMMON_IND v_NB_SRB_COMMON_IND;

    var template NAS_UL_Message_Type v_NAS_UL_Message := omit;

f_NBIOT_SS_CommonCellConfig(p_CellId, cads_NB_ULGrantScheduling_REQ(p_CellId, cs_TimingInfo_Now, false, -, cs_NB_UL_GrantScheduling_Start));
    SRB.receive(car_NB_SRB0_RrcPdu_IND(p_CellId, cr_RRCConnectionResumeRequest_NB(p_ResumeIdentity, p_ShortMAC_I, p_ResumeCause)));

    v_SecurityParams := f_NBIOT_Security_Get();

    v_SecurityParams := f_NBIOT_SS_AS_ReestablishResumeSecurity(p_CellId, v_SecurityParams, rrcConnectionResume);

    f_NBIOT_Security_Set(v_SecurityParams);

    SRB.send(cas_NB_SRB_RrcPdu_REQ(p_CellId, tsc_SRB1, cs_TimingInfo_Now, cs_RRCConnectionResume_NB(v_RRC_TI, v_RadioResourceConfigDedicated)));

    alt {

      [] SRB.receive(car_NB_SRB_RrcNasPdu_IND(p_CellId, tsc_SRB1, cr_RRCConnectionResumeComplete_NB(v_RRC_TI), cr_NAS_Indication(p_ExpectedSecurityStatus, ?))) -> value v_NB_SRB_COMMON_IND {

        v_NAS_UL_Message := v_NB_SRB_COMMON_IND.Signalling.Nas[0].Pdu.Msg;

        if (not match(valueof(v_NAS_UL_Message), p_Expected_NAS_UL_Message)) {

          f_NBIOT_SetVerdictFailOrInconc (__FILE__, __LINE__, "RRCConnectionResumeComplete: Piggybacked NAS message does not match");

        }

        //return v_NAS_UL_Message;
      }

      [] SRB.receive(car_NB_SRB_RrcPdu_IND(p_CellId, tsc_SRB1, cr_RRCConnectionResumeComplete_NB(v_RRC_TI))) {

        if (ispresent(p_Expected_NAS_UL_Message)) {

          f_NBIOT_SetVerdictFailOrInconc (__FILE__, __LINE__, "RRCConnectionResumeComplete: Piggybacked NAS message expected");

        }

      }

    }

    f_NBIOT_SS_ConfigRachProcedureMsg4(p_CellId, f_NBIOT_508_RRCConnectionSetup());

    return v_NAS_UL_Message;
  }
2.4 Change 4 – Correction to function ‘function f_NBIOT_TrackingAreaUpdate_Accept_Complete()’

	Function name
	f_NBIOT_TrackingAreaUpdate_Accept_Complete()

	Reason for change
	1. Add new parameter p_ResumeFlag for procedure 8.1.5A.6 (UE resume RRC connection on a new NB-IOT cell) and 8.1.5A.5 (NB-IoT UE is camped on a new NB-IOT cell).
2.  Add new variable v_SrbId instead of SRB1bis.

	Summary of change
	1.  Add new parameter p_ResumeFlag for procedure 8.1.5A.6 and 8.1.5A.5.
2.  Add new variable v_SrbId instead of SRB1bis.

	TTCN module
	NBIOT_NASSteps.ttcn

	MCC160 Comment
	


Before change

  function f_NBIOT_TrackingAreaUpdate_Accept_Complete(NBIOT_CellId_Type                       p_CellId,

                                                      IOT_STATE_Type                          p_IOT_State,

                                                      GutiParameters_Type                     p_GutiParams,

                                                      template (omit) EPS_BearerContextStatus p_EPS_ContextStatus,

                                                      template (omit) AdditionalUpdateType    p_AdditionalUpdateType, // @sic R5s170351 sic@

                                                      template (omit) PLMN_List               p_EPLMNlist := omit) runs on NBIOT_PTC

  {

    var NAS_PlmnId v_PLMN := f_Asn2Nas_PlmnId(p_GutiParams.PLMN_Identity);

    var EUTRA_ASN1_TrackingAreaCode_Type v_Tac := f_NBIOT_CellInfo_GetTAC(p_CellId);

    var template (value) TrackingAreaIdList v_TaiList := cds_TAIListNonConsecutive_tlv(v_PLMN, { bit2oct(v_Tac) });

    /* Transmit TRACKING AREA UPDATE ACCEPT message */

    /* In case MME has the cached security context it verifies the TAU Request message

       and replies with TAU Accept message protected with the cached security context

       ( 33.401 cl. 9.1.2 Idle mode procedures in E-UTRAN ) */

    SRB.send(cas_NB_SRB_NasPdu_REQ(p_CellId,

                                   tsc_SRB1bis,

                                   cs_TimingInfo_Now,

                                   cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                  cs_TAU_Accept_Common(tsc_EpsUpdate_TaUpdate,

                                                                       f_GutiParameters2MobileIdentity (tsc_IEI_Guti, p_GutiParams),

                                                                       v_TaiList,

                                                                       p_EPS_ContextStatus,

                                                                       omit,

                                                                       omit,

                                                                       f_GetNBIOTAdditionalUpdateResult(p_AdditionalUpdateType, p_IOT_State),

                                                                       -,

                                                                       -,

                                                                       p_EPLMNlist,

                                                                       f_GetNBIOTNetworkFeatureSupport(p_IOT_State)))));

    /* Receive TRACKING AREA UPDATE COMPLETE */

    /* According to 24.301 cl: 5.5.3.2.4:

       If the TRACKING AREA UPDATE ACCEPT message contained a GUTI,

       the UE shall return a TRACKING AREA UPDATE COMPLETE message to

       the MME to acknowledge the received GUTI */

    SRB.receive(car_NB_SRB_NasPdu_IND(p_CellId,

                                      tsc_SRB1bis,

                                      cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                        cr_508_TAU_Complete)));

  }
After change

  function f_NBIOT_TrackingAreaUpdate_Accept_Complete(NBIOT_CellId_Type                       p_CellId,

                                                      IOT_STATE_Type                          p_IOT_State,

                                                      GutiParameters_Type                     p_GutiParams,

                                                      template (omit) EPS_BearerContextStatus p_EPS_ContextStatus,

                                                      template (omit) AdditionalUpdateType    p_AdditionalUpdateType, // @sic R5s170351 sic@

                                                      template (omit) PLMN_List               p_EPLMNlist := omit,












                   boolean                                 p_ResumeFlag := false) runs on NBIOT_PTC

  {

    var NAS_PlmnId v_PLMN := f_Asn2Nas_PlmnId(p_GutiParams.PLMN_Identity);

    var EUTRA_ASN1_TrackingAreaCode_Type v_Tac := f_NBIOT_CellInfo_GetTAC(p_CellId);

    var template (value) TrackingAreaIdList v_TaiList := cds_TAIListNonConsecutive_tlv(v_PLMN, { bit2oct(v_Tac) });

    var NB_SRB_Identity_Type v_SrbId := tsc_SRB1bis;
    /* Transmit TRACKING AREA UPDATE ACCEPT message */

    /* In case MME has the cached security context it verifies the TAU Request message

       and replies with TAU Accept message protected with the cached security context

       ( 33.401 cl. 9.1.2 Idle mode procedures in E-UTRAN ) */


if(p_ResumeFlag){


  v_SrbId := tsc_SRB1;


}

    SRB.send(cas_NB_SRB_NasPdu_REQ(p_CellId,

                                   v_SrbId,

                                   cs_TimingInfo_Now,

                                   cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                  cs_TAU_Accept_Common(tsc_EpsUpdate_TaUpdate,

                                                                       f_GutiParameters2MobileIdentity (tsc_IEI_Guti, p_GutiParams),

                                                                       v_TaiList,

                                                                       p_EPS_ContextStatus,

                                                                       omit,

                                                                       omit,

                                                                       f_GetNBIOTAdditionalUpdateResult(p_AdditionalUpdateType, p_IOT_State),

                                                                       -,

                                                                       -,

                                                                       p_EPLMNlist,

                                                                       f_GetNBIOTNetworkFeatureSupport(p_IOT_State)))));

    /* Receive TRACKING AREA UPDATE COMPLETE */

    /* According to 24.301 cl: 5.5.3.2.4:

       If the TRACKING AREA UPDATE ACCEPT message contained a GUTI,

       the UE shall return a TRACKING AREA UPDATE COMPLETE message to

       the MME to acknowledge the received GUTI */

    SRB.receive(car_NB_SRB_NasPdu_IND(p_CellId,

                                      v_SrbId,

                                      cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                        cr_508_TAU_Complete)));

  }
2.5 Change 5 – Correction to function ‘function f_NBIOT_508CheckRRCConnectionResumeOnNewCell()’

	Function name
	f_NBIOT_508CheckRRCConnectionResumeOnNewCell()

	Reason for change
	Since new parameter p_ExpectedSecurityStatus is added in f_NBIOT_RRCConnectionResume() and new parameter p_ResumeFlag is added in f_NBIOT_TrackingAreaUpdate_Accept_Complete(), accordingly add tsc_SHT_IntegrityProtected for TAU REQ message and Resume flag = true.

	Summary of change
	Accordingly add tsc_SHT_IntegrityProtected for TAU REQ message and Resume flag = true.

	TTCN module
	NBIOT_NASSteps.ttcn

	MCC160 Comment
	


Before change

  function f_NBIOT_508CheckRRCConnectionResumeOnNewCell(NBIOT_CellId_Type p_CellId,

                                                        template (present) EUTRA_ASN1_ResumeIdentity_r13_Type p_ResumeIdentity := ?,

                                                        template (present) EUTRA_ASN1_ShortMAC_I_Type p_ShortMAC_I := ?,

                                                        template (present) EstablishmentCause_NB_r13 p_ResumeCause := ?) runs on NBIOT_PTC

  { /* Procedure according to 8.1.5A.6 */

    var NAS_UL_Message_Type v_NasMsg;
    var GutiParameters_Type v_GutiParams := f_NBIOT_CellInfo_GetGuti(p_CellId);

    var NAS_KsiValue v_KsiValue := f_NBIOT_SecurityKSIasme_Get();

    var template (present) NAS_UL_Message_Type v_ExpectedTAU_Request;

    var template (omit) EPS_BearerContextStatus v_EPS_ContextStatus;

    var AdditionalUpdateType v_AdditionalUpdateType;

    v_ExpectedTAU_Request := cr_508_TAU_Request(tsc_EpsUpdate_TaUpdate,

                                                v_KsiValue,

                                                cr_DRXparameter_Any ('5C'O) ifpresent,

                                                cr_AdditionalUpdateType_CIOT ifpresent);

    /* steps 1 - 3 */

    v_NasMsg := valueof(f_NBIOT_RRCConnectionResume(p_CellId, p_ResumeIdentity, p_ShortMAC_I, p_ResumeCause, v_ExpectedTAU_Request));

    v_EPS_ContextStatus := v_NasMsg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus; // Send this back in the Accept

    v_AdditionalUpdateType := v_NasMsg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType;

    /* steps 4 - 5 */

    f_NBIOT_TrackingAreaUpdate_Accept_Complete(p_CellId, USER_PLANE, v_GutiParams, v_EPS_ContextStatus, v_AdditionalUpdateType);

    /* Step 6: Release RRC connection */

    f_NBIOT_RRC_ConnectionRelease(p_CellId, USER_PLANE, tsc_RRC_TI_Def);

  }
After change

  function f_NBIOT_508CheckRRCConnectionResumeOnNewCell(NBIOT_CellId_Type p_CellId,

                                                        template (present) EUTRA_ASN1_ResumeIdentity_r13_Type p_ResumeIdentity := ?,

                                                        template (present) EUTRA_ASN1_ShortMAC_I_Type p_ShortMAC_I := ?,

                                                        template (present) EstablishmentCause_NB_r13 p_ResumeCause := ?) runs on NBIOT_PTC

  { /* Procedure according to 8.1.5A.6 */

    var template NAS_UL_Message_Type v_NasMsg;

    var GutiParameters_Type v_GutiParams := f_NBIOT_CellInfo_GetGuti(p_CellId);

    var NAS_KsiValue v_KsiValue := f_NBIOT_SecurityKSIasme_Get();

    var template (present) NAS_UL_Message_Type v_ExpectedTAU_Request;

    var template (omit) EPS_BearerContextStatus v_EPS_ContextStatus;

    var template AdditionalUpdateType v_AdditionalUpdateType;
    v_ExpectedTAU_Request := cr_508_TAU_Request(tsc_EpsUpdate_TaUpdate,

                                                v_KsiValue,

                                                cr_DRXparameter_Any ('5C'O) ifpresent,

                                                cr_AdditionalUpdateType_CIOT ifpresent);

    /* steps 1 - 3 */

    v_NasMsg := f_NBIOT_RRCConnectionResume(p_CellId, p_ResumeIdentity, p_ShortMAC_I, p_ResumeCause, v_ExpectedTAU_Request, tsc_SHT_IntegrityProtected);//For TAU REQ
    v_EPS_ContextStatus := v_NasMsg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus; // Send this back in the Accept

    v_AdditionalUpdateType := v_NasMsg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType;

    /* steps 4 - 5 */

    f_NBIOT_TrackingAreaUpdate_Accept_Complete(p_CellId, USER_PLANE, v_GutiParams, v_EPS_ContextStatus, v_AdditionalUpdateType, -, true);//Resume flag = true
    /* Step 6: Release RRC connection */

    f_NBIOT_RRC_ConnectionRelease(p_CellId, USER_PLANE, tsc_RRC_TI_Def);

  }
2.6 Change 6 – Correction to function ‘function f_EUTRA_NB_VarShortMAC()’

	Function name
	f_EUTRA_NB_VarShortMAC ()

	Reason for change
	1.  Add new parameter p_ResumeFlag.
2.  Modify v_Bitstring_VarShortMAC_Input for RRC Resume Procedure as VarShortResumeMAC-Input-NB-r13 according to TS 36.331.
VarShortResumeMAC-Input-NB-r13
::= 
VarShortResumeMAC-Input-r13
VarShortResumeMAC-Input-r13 ::=

SEQUENCE {


cellIdentity-r13





CellIdentity,


physCellId-r13






PhysCellId,


c-RNTI-r13







C-RNTI,


resumeDiscriminator-r13




BIT STRING(SIZE(1))

}

	Summary of change
	1.  Add new parameter p_ResumeFlag.

2.  Modify v_Bitstring_VarShortMAC_Input for RRC Resume Procedure as VarShortResumeMAC-Input-NB-r13 according to TS 36.331.

	TTCN module
	EUTRA_NB_SecurityFunctions.ttcn

	MCC160 Comment
	


Before change

  function f_EUTRA_NB_VarShortMAC(VarShortMAC_Input_Type p_VarShortMAC_Input) return octetstring

  {

    var bitstring v_PhysCellId;

    var bitstring v_Bitstring_VarShortMAC_Input;

    var octetstring v_Octetstring_VarShortMAC_Input;

    v_PhysCellId := int2bit(p_VarShortMAC_Input.physCellId, 9);

    //Place IEs of VarShortMAC_Input into single bitstring

    v_Bitstring_VarShortMAC_Input := p_VarShortMAC_Input.cellIdentity & v_PhysCellId & p_VarShortMAC_Input.c_RNTI;

    //Octet align the bit string

    v_Bitstring_VarShortMAC_Input := f_OctetAlignedBitString(v_Bitstring_VarShortMAC_Input);  // @sic R5s100324 sic@

    //Convert bitstring to octet aligned octetstring

    v_Octetstring_VarShortMAC_Input := bit2oct(v_Bitstring_VarShortMAC_Input);

    return (v_Octetstring_VarShortMAC_Input);

  }
After change

  function f_EUTRA_NB_VarShortMAC(VarShortMAC_Input_Type p_VarShortMAC_Input, 

                                  boolean p_ResumeFlag := false) return octetstring

  {
    var bitstring v_PhysCellId;

    var bitstring v_Bitstring_VarShortMAC_Input;

    var octetstring v_Octetstring_VarShortMAC_Input;

    v_PhysCellId := int2bit(p_VarShortMAC_Input.physCellId, 9);

    //Place IEs of VarShortMAC_Input into single bitstring

v_Bitstring_VarShortMAC_Input := p_VarShortMAC_Input.cellIdentity & v_PhysCellId & p_VarShortMAC_Input.c_RNTI;

if( p_ResumeFlag == true)    


{



v_Bitstring_VarShortMAC_Input := v_Bitstring_VarShortMAC_Input&'1'B;


}
    //Octet align the bit string

    v_Bitstring_VarShortMAC_Input := f_OctetAlignedBitString(v_Bitstring_VarShortMAC_Input);  // @sic R5s100324 sic@

    //Convert bitstring to octet aligned octetstring

    v_Octetstring_VarShortMAC_Input := bit2oct(v_Bitstring_VarShortMAC_Input);

    return (v_Octetstring_VarShortMAC_Input);

  }
2.7 Change 7 – Correction to function ‘function f_NBIOT_Calculate_ShortMAC()’

	Function name
	f_NBIOT_Calculate_ShortMAC()

	Reason for change
	Add new parameter p_ResumeFlag for VarShortResumeMAC-Input-NB-r13.


	Summary of change
	Add new parameter p_ResumeFlag for VarShortResumeMAC-Input-NB-r13.

	TTCN module
	NBIOT_SecuritySteps.ttcn

	MCC160 Comment
	


Before change

  function f_NBIOT_Calculate_ShortMAC(NBIOT_CellId_Type p_CellId_Source,

                                      template (omit) NBIOT_CellId_Type p_CellId_Current := omit) runs on NBIOT_PTC return EUTRA_ASN1_ShortMAC_I_Type

  {

    var NBIOT_CellId_Type v_CellId_Current := p_CellId_Source;

    var EUTRA_ASN1_ShortMAC_I_Type v_ShortMAC_I;

    var O4_Type v_CalculatedMac;

    var NBIOT_SecurityParams_Type v_SecurityParams := f_NBIOT_Security_Get();

    var VarShortMAC_Input_Type v_VarShortMAC_Input;

    var octetstring v_Encoded_VarShortMAC_Input;

    var O4_Type v_PdcpCount := 'FFFFFFFF'O;

    var B5_Type v_BearerId := '11111'B;

    var integer v_Direction := 1;

    if (ispresent(p_CellId_Current)) {

      v_CellId_Current := valueof(p_CellId_Current);

    }

    v_VarShortMAC_Input.cellIdentity := f_NBIOT_CellInfo_GetCellIdentity(v_CellId_Current);/* Acc. 36.331 ch. 7.1 Set to CellIdentity of the current cell. */

    v_VarShortMAC_Input.physCellId := f_NBIOT_CellInfo_GetPhyCellId(p_CellId_Source);      /* Set to the physical cell identity of the cell the UE was connected to prior to the failure. */

    v_VarShortMAC_Input.c_RNTI := f_NBIOT_CellInfo_GetC_RNTI(p_CellId_Source);             /* Set to C-RNTI that the UE had in the cell it was connected to prior to the failure. */

    /* According to 36.331 cl. 5.3.7.4

       2>  set the shortMAC-I to the 16 least significant bits of the MAC-I calculated:

       3>  over the ASN.1 encoded VarShortMAC-Input;

       3>  with the KRRCint key and integrity protection algorithm that was used in the

       source cell (handover and mobility from E-UTRA failure) or of the cell in which

       the trigger for the re-establishment occurred (other cases); and

       3>  with all input bits for COUNT, BEARER and DIRECTION set to binary ones; */

    v_Encoded_VarShortMAC_Input := f_EUTRA_NB_VarShortMAC(v_VarShortMAC_Input);

    v_CalculatedMac := fx_AsIntegrityAlgorithm(v_Encoded_VarShortMAC_Input,

                                               v_SecurityParams.AS_Integrity.Algorithm,

                                               v_SecurityParams.AS_Integrity.KRRCint,

                                               v_PdcpCount,

                                               v_BearerId,

                                               v_Direction);

    v_ShortMAC_I := substr(oct2bit(v_CalculatedMac),16,16); /* According to 36.331 ch. 5.3.7.4 2 Set the shortMAC-I to the 16 least significant bits of the MAC-I calculated */

    return v_ShortMAC_I;

  }
After change

  function f_NBIOT_Calculate_ShortMAC(NBIOT_CellId_Type p_CellId_Source,

                                      template (omit) NBIOT_CellId_Type p_CellId_Current := omit,










         boolean p_ResumeFlag := false) runs on NBIOT_PTC return EUTRA_ASN1_ShortMAC_I_Type

  {

    var NBIOT_CellId_Type v_CellId_Current := p_CellId_Source;

    var EUTRA_ASN1_ShortMAC_I_Type v_ShortMAC_I;

    var O4_Type v_CalculatedMac;

    var NBIOT_SecurityParams_Type v_SecurityParams := f_NBIOT_Security_Get();

    var VarShortMAC_Input_Type v_VarShortMAC_Input;

    var octetstring v_Encoded_VarShortMAC_Input;

    var O4_Type v_PdcpCount := 'FFFFFFFF'O;

    var B5_Type v_BearerId := '11111'B;

    var integer v_Direction := 1;

    if (ispresent(p_CellId_Current)) {

      v_CellId_Current := valueof(p_CellId_Current);

    }

    v_VarShortMAC_Input.cellIdentity := f_NBIOT_CellInfo_GetCellIdentity(v_CellId_Current);/* Acc. 36.331 ch. 7.1 Set to CellIdentity of the current cell. */

    v_VarShortMAC_Input.physCellId := f_NBIOT_CellInfo_GetPhyCellId(p_CellId_Source);      /* Set to the physical cell identity of the cell the UE was connected to prior to the failure. */

    v_VarShortMAC_Input.c_RNTI := f_NBIOT_CellInfo_GetC_RNTI(p_CellId_Source);             /* Set to C-RNTI that the UE had in the cell it was connected to prior to the failure. */

    /* According to 36.331 cl. 5.3.7.4

       2>  set the shortMAC-I to the 16 least significant bits of the MAC-I calculated:

       3>  over the ASN.1 encoded VarShortMAC-Input;

       3>  with the KRRCint key and integrity protection algorithm that was used in the

       source cell (handover and mobility from E-UTRA failure) or of the cell in which

       the trigger for the re-establishment occurred (other cases); and

       3>  with all input bits for COUNT, BEARER and DIRECTION set to binary ones; */

    v_Encoded_VarShortMAC_Input := f_EUTRA_NB_VarShortMAC(v_VarShortMAC_Input, p_ResumeFlag);

    v_CalculatedMac := fx_AsIntegrityAlgorithm(v_Encoded_VarShortMAC_Input,

                                               v_SecurityParams.AS_Integrity.Algorithm,

                                               v_SecurityParams.AS_Integrity.KRRCint,

                                               v_PdcpCount,

                                               v_BearerId,

                                               v_Direction);

    v_ShortMAC_I := substr(oct2bit(v_CalculatedMac),16,16); /* According to 36.331 ch. 5.3.7.4 2 Set the shortMAC-I to the 16 least significant bits of the MAC-I calculated */

    return v_ShortMAC_I;

  }
2.8 Change 8 – Correction to function ‘function f_TC_22_4_15_NBIOT()’

	Function name
	f_TC_22_4_15_NBIOT ()

	Reason for change
	1． Incorrectly setting f_NBIOT_SetCellPowerList for T0 before activating Cell1 will cause error. Change the position of or delete superfluous  f_NBIOT_SetCellPowerList for T0.
2． Incorrect v_ShortMAC_I calculation.
3． Incorrect Msg4 configuration of cell 11 before step 8.
4． Incorrect postamble state.

	Summary of change
	1． Change the position of or delete superfluous f_NBIOT_SetCellPowerList for T0.
2． Correct v_ShortMAC_I calculation.
3． Correct Msg4 configuration of cell 11 to none before step 8.
4． Correct postamble state.

	TTCN module
	NBIOT_RRC_UserPlane.ttcn

	MCC160 Comment
	


Before change

  function f_TC_22_4_15_NBIOT() runs on NBIOT_PTC

  {

    var EUTRA_ASN1_ResumeIdentity_r13_Type v_ResumeIdentityStep2 := tsc_NBIOT_ResumeIdentity_Def;

    var EUTRA_ASN1_ResumeIdentity_r13_Type v_ResumeIdentityStep7 := not4b v_ResumeIdentityStep2;

    var EUTRA_ASN1_ShortMAC_I_Type v_ShortMAC_I;

    var template (value) NB_CellPowerList_Type v_CellPowerList_AtT0 := {

      cs_NB_CellPower(nbiot_Cell1, -85),

      cs_NB_CellPower(nbiot_Cell11, tsc_NonSuitableOffNBIOTCellRS_EPRE)

    };

    var template (value) NB_CellPowerList_Type v_CellPowerList_AtT1 := {

      cs_NB_CellPower(nbiot_Cell1, -85),

      cs_NB_CellPower(nbiot_Cell11, -73)

    };

    f_NBIOT_Init(c3, USER_PLANE);

    f_NBIOT_CellInfo_InitMaxReferencePower(nbiot_Cell11, -73);

    f_NBIOT_SetCellPowerList(v_CellPowerList_AtT0);
    // Create and configure all cells

    f_NBIOT_CellConfig_Def(nbiot_Cell1, USER_PLANE);

    f_NBIOT_CellConfig_Def(nbiot_Cell11, USER_PLANE);

    f_NBIOT_Preamble(nbiot_Cell1, USER_PLANE);

    f_NBIOT_UE_Detach_SwitchOffUe(nbiot_Cell1, USER_PLANE, RRC_CONNECTED);

    f_NBIOT_TestBody_Set(true);

    //@siclog "Step 1" siclog@

    f_NBIOT_Preamble(nbiot_Cell1, USER_PLANE, STATE2_NB_CONNECTED_UPDATED);

    //@siclog "Step 2" siclog@

    f_NBIOT_RRCConnectionSuspend(nbiot_Cell1, v_ResumeIdentityStep2);

    v_ShortMAC_I := f_NBIOT_Calculate_ShortMAC(nbiot_Cell1);

    //@siclog "Step 3" siclog@

    f_NBIOT_UE_Page(nbiot_Cell1);

    //@siclog "Step 4 - 6" siclog@

    f_NBIOT_RRCConnectionResume(nbiot_Cell1, v_ResumeIdentityStep2, v_ShortMAC_I, mt_Access, omit);

    //@siclog "Step 7" siclog@

    f_NBIOT_RRCConnectionSuspend(nbiot_Cell1, v_ResumeIdentityStep7);

    //@siclog "Step 8" siclog@

    f_NBIOT_SetCellPowerList(v_CellPowerList_AtT1);

    //@siclog "Step 9" siclog@

    v_ShortMAC_I := f_NBIOT_Calculate_ShortMAC(nbiot_Cell1, nbiot_Cell11);

    f_NBIOT_508CheckRRCConnectionResumeOnNewCell(nbiot_Cell11, v_ResumeIdentityStep7, v_ShortMAC_I, mo_Signalling);

    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, " Step 9");

    f_NBIOT_TestBody_Set(false);

    //Switch/power off UE

    f_NBIOT_Postamble(nbiot_Cell11, USER_PLANE, N2_CONNECTED);

  }
After change

  function f_TC_22_4_15_NBIOT() runs on NBIOT_PTC

  {

    var EUTRA_ASN1_ResumeIdentity_r13_Type v_ResumeIdentityStep2 := tsc_NBIOT_ResumeIdentity_Def;

    var EUTRA_ASN1_ResumeIdentity_r13_Type v_ResumeIdentityStep7 := not4b v_ResumeIdentityStep2;

    var EUTRA_ASN1_ShortMAC_I_Type v_ShortMAC_I;

    var template (value) NB_CellPowerList_Type v_CellPowerList_AtT0 := {

      cs_NB_CellPower(nbiot_Cell1, -85),

      cs_NB_CellPower(nbiot_Cell11, tsc_NonSuitableOffNBIOTCellRS_EPRE)

    };

    var template (value) NB_CellPowerList_Type v_CellPowerList_AtT1 := {

      cs_NB_CellPower(nbiot_Cell1, -85),

      cs_NB_CellPower(nbiot_Cell11, -73)

    };

    f_NBIOT_Init(c3, USER_PLANE);

    f_NBIOT_CellInfo_InitMaxReferencePower(nbiot_Cell11, -73);

    // Create and configure all cells

    f_NBIOT_CellConfig_Def(nbiot_Cell1, USER_PLANE);

    f_NBIOT_CellConfig_Def(nbiot_Cell11, USER_PLANE);

    f_NBIOT_Preamble(nbiot_Cell1, USER_PLANE);

    f_NBIOT_UE_Detach_SwitchOffUe(nbiot_Cell1, USER_PLANE, RRC_CONNECTED);

    f_NBIOT_TestBody_Set(true);

    f_NBIOT_SetCellPowerList(v_CellPowerList_AtT0);
    //@siclog "Step 1" siclog@

    f_NBIOT_Preamble(nbiot_Cell1, USER_PLANE, STATE2_NB_CONNECTED_UPDATED);

    //@siclog "Step 2" siclog@

    f_NBIOT_RRCConnectionSuspend(nbiot_Cell1, v_ResumeIdentityStep2);

    v_ShortMAC_I := f_NBIOT_Calculate_ShortMAC(nbiot_Cell1, -, true);

    //@siclog "Step 3" siclog@

    f_NBIOT_UE_Page(nbiot_Cell1);

    //@siclog "Step 4 - 6" siclog@

    f_NBIOT_RRCConnectionResume(nbiot_Cell1, v_ResumeIdentityStep2, v_ShortMAC_I, mt_Access, omit);

    //@siclog "Step 7" siclog@

    f_NBIOT_RRCConnectionSuspend(nbiot_Cell1, v_ResumeIdentityStep7);

    f_NBIOT_SS_ConfigRachProcedureMsg4(nbiot_Cell11);//set msg 4 to None for cell11
    //@siclog "Step 8" siclog@

    f_NBIOT_SetCellPowerList(v_CellPowerList_AtT1);

    //@siclog "Step 9" siclog@

    v_ShortMAC_I := f_NBIOT_Calculate_ShortMAC(nbiot_Cell1, nbiot_Cell11, true);
    f_NBIOT_508CheckRRCConnectionResumeOnNewCell(nbiot_Cell11, v_ResumeIdentityStep7, v_ShortMAC_I, mo_Signalling);

    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, " Step 9");

    f_NBIOT_TestBody_Set(false);

    //Switch/power off UE

    f_NBIOT_Postamble(nbiot_Cell11, USER_PLANE, N1_IDLE);

  }
3. Branches executed

None.

4. Execution Log Files

4.1 RDA8909
The RDA8909 passed this test case on Starpoint SP8630 Integration Test System in LTE FDD band 8. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file
TC_22_4_15_RDA8909_PASS.spm

In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

· PICS/PIXIT file name
TC_22_4_15.sppar

5. References

	[1]
	R5s170516: Supporting information for agreement of NB-IoT test case 22.4.15. This archive comprises html and text format execution log files and PICS/PIXIT settings file respectively.


